Pensieve header: The main lemma for ${\mathfrak g}_1$ using gl_2.

sl(2)

```
rh = (1  0
      0 -1); re = (0 1
                   0 0); rf = (0 0
                               1 0);
MB[x_?MatrixQ, y_?MatrixQ] := x.y - y.x;
```

```
MB[rh, re] // MatrixForm
```

$\begin{pmatrix} 0 & 2 \\ 0 & 0 \end{pmatrix}$

```
MB[rh, rf] // MatrixForm
```

$\begin{pmatrix} 0 & 0 \\ -2 & 0 \end{pmatrix}$

```
MB[re, rf] // MatrixForm
```

$\begin{pmatrix} 1 & 0 \\ 0 & -1 \end{pmatrix}$

```
MatrixExp[α re]
```

{{1, α}, {0, 1}}

```
MatrixExp[α re]
```

{{1, 0}, {β, 1}}

```
MatrixExp[α rf].MatrixExp[β re] // MatrixForm
```

$\begin{pmatrix} 1 & \beta \\ \alpha & 1+\alpha\beta \end{pmatrix}$

```
MatrixExp[a re].MatrixExp[b rf].MatrixExp[c rh] // MatrixForm
```

$\begin{pmatrix} (1+a b)\, e^{c} & a\, e^{-c} \\ b\, e^{c} & e^{-c} \end{pmatrix}$

```
eqn2 = MatrixExp[α rf].MatrixExp[β re] == MatrixExp[a re].MatrixExp[b rf].MatrixExp[c rh]
```

$\{\{1, \beta\}, \{\alpha, 1+\alpha\beta\}\} == \{\{(1+a b)\, e^{c}, a\, e^{-c}\}, \{b\, e^{c}, e^{-c}\}\}$

```
Flatten /@ eqn2
```

$\{1, \beta, \alpha, 1+\alpha\beta\} == \{(1+a b)\, e^{c}, a\, e^{-c}, b\, e^{c}, e^{-c}\}$

```
eqns3 = Thread[Flatten /@ eqn2]
```

$\{1 == (1+a b)\, e^{c}, \beta == a\, e^{-c}, \alpha == b\, e^{c}, 1+\alpha\beta == e^{-c}\}$

```
Solve[eqns3, {a, b, c}]
```

$\{\{a \to \frac{\beta}{1+\alpha\beta}, b \to \alpha\,(1+\alpha\beta), c \to \text{ConditionalExpression}[2\, i\, \pi\, C[1] + \text{Log}[\frac{1}{1+\alpha\beta}], C[1] \in \text{Integers}]\}\}$

```
Solve[e^x == 1, x]
```

$\{\{x \to \text{ConditionalExpression}[2\, i\, \pi\, C[1], C[1] \in \text{Integers}]\}\}$

```
Solve[eqns3, {a, b, c}] /. C[1] → 0
```

$\{\{a \to \frac{\beta}{1+\alpha\beta}, b \to \alpha\,(1+\alpha\beta), c \to \text{Log}[\frac{1}{1+\alpha\beta}]\}\}$

Representing $\mathfrak{g}^\epsilon = \langle h, e, l, f\rangle / ([e, l] = -e,\ [f, l] = f,\ [e, f] = h - 2\epsilon l,\ [h, *] = 0)$

```
ρh = (-1 0; 0 -1); ρe = (0 0; -ϵ 0); ρl = (-(1+1/ϵ)/2 0; 0 (1-1/ϵ)/2); ρf = (0 1; 0 0);
MB[x_?MatrixQ, y_?MatrixQ] := x.y - y.x;
Simplify@{MB[ρe, ρl] == -ρe, MB[ρf, ρl] == ρf, MB[ρe, ρf] == ρh - 2 ϵ ρl}
```

```
{True, True, True}
```

Some $\mathfrak{g}^\epsilon$ lemmas

1. $\mathbb{O}\left(e^{\gamma l+\beta e} \mid l e\right) = \mathbb{O}\left(e^{\gamma l+e^\gamma \beta e} \mid e l\right)$.
2. $\mathbb{O}\left(e^{\gamma l+\beta f} \mid f l\right) = \mathbb{O}\left(e^{\gamma l+e^\gamma \beta f} \mid l f\right)$.

Proofs.

```
MatrixForm /@ {MatrixExp[γ ρl].MatrixExp[β ρe], MatrixExp[e^γ β ρe].MatrixExp[γ ρl]}
```

$$\left\{\begin{pmatrix} e^{-\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} & 0 \\ -e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}}\beta\epsilon & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} \end{pmatrix}, \begin{pmatrix} e^{-\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} & 0 \\ -e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}}\beta\epsilon & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} \end{pmatrix}\right\}$$

```
MatrixForm /@ {MatrixExp[β ρf].MatrixExp[γ ρl], MatrixExp[γ ρl].MatrixExp[e^γ β ρf]}
```

$$\left\{\begin{pmatrix} e^{-\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}}\beta \\ 0 & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} \end{pmatrix}, \begin{pmatrix} e^{-\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}}\beta \\ 0 & e^{\frac{\gamma}{2}-\frac{\gamma}{2\epsilon}} \end{pmatrix}\right\}$$

Deriving “lemma 3 at $\delta = 0$”:

```
MatrixExp[α ρf].MatrixExp[β ρe] // Simplify // MatrixForm
```

$$\begin{pmatrix} 1-\alpha\beta\epsilon & \alpha \\ -\beta\epsilon & 1 \end{pmatrix}$$

```
eqn = MatrixExp[α ρf].MatrixExp[β ρe] == MatrixExp[a ρe].MatrixExp[c (ρh - 2 ϵ ρl)].MatrixExp[b ρf]
```

$$\{\{1-\alpha\beta\epsilon, \alpha\}, \{-\beta\epsilon, 1\}\} == \left\{\left\{e^{c\epsilon}, b\, e^{c\epsilon}\right\}, \left\{-a\, e^{c\epsilon}\epsilon, e^{-c\epsilon} - a\, b\, e^{c\epsilon}\epsilon\right\}\right\}$$

```
sol = Solve[Thread[Flatten /@ eqn], {a, b, c}][[1]]
```

$$\left\{a \to -\frac{\beta}{-1+\alpha\beta\epsilon}, b \to -\frac{\alpha}{-1+\alpha\beta\epsilon}, c \to \text{ConditionalExpression}\left[\frac{2 i \pi C[1] + \text{Log}[1-\alpha\beta\epsilon]}{\epsilon}, C[1] \in \text{Integers}\right]\right\}$$

```
sol = sol /. C[1] → 0
```

$$\left\{a \to -\frac{\beta}{-1+\alpha\beta\epsilon}, b \to -\frac{\alpha}{-1+\alpha\beta\epsilon}, c \to \frac{\text{Log}[1-\alpha\beta\epsilon]}{\epsilon}\right\}$$

And so,

3. $\mathbb{O}\left(e^{\alpha f+\beta e} \mid f e\right) = \mathbb{O}\left(e^{c h + a e - 2\epsilon c l + b f} \mid e l f\right)$, with $\left\{a \to -\frac{\beta}{-1+\alpha\beta\epsilon}, b \to -\frac{\alpha}{-1+\alpha\beta\epsilon}, c \to \frac{\text{Log}[1-\alpha\beta\epsilon]}{\epsilon}\right\}$.

At $\epsilon = 0$ get

```
Limit[{a, b, c} /. sol, ϵ → 0]
```

$$\{\beta, \alpha, -\alpha\beta\}$$

The full $\mathfrak{g}_1$ lemma

At $\epsilon^2 = 0$ get

```
Series[{a, b, c} /. sol, {ϵ, 0, 1}] // Normal
```

$$\left\{\beta + \alpha\beta^2\epsilon, \alpha + \alpha^2\beta\epsilon, -\alpha\beta - \frac{1}{2}\alpha^2\beta^2\epsilon\right\}$$

```
Collect[Series[ⅇ^(c h + a e - 2 ϵ c l + b f) /. sol, {ϵ, 0, 1}] // Normal, ⅇ-, Expand]
```

$e^{f\alpha+e\beta-h\alpha\beta}\left(1+2\,l\,\alpha\,\beta\,\epsilon+f\,\alpha^2\,\beta\,\epsilon+e\,\alpha\,\beta^2\,\epsilon-\frac{1}{2}\,h\,\alpha^2\,\beta^2\,\epsilon\right)$

Which means $\mathbb{O}\left(e^{\alpha f+\beta e}\,\middle|\,f\,e\right)=\mathbb{O}\left(e^{-\alpha\beta h+\beta e+\alpha f}\left(1+\epsilon\left(2\,l\,\alpha\,\beta+f\,\alpha^2\,\beta+e\,\alpha\,\beta^2-\frac{1}{2}\,h\,\alpha^2\,\beta^2\right)\right)\,\middle|\,e\,l\,f\right)$.

And so,

3. $\mathbb{O}\left(e^{\alpha f+\beta e+\delta e f}\,\middle|\,f\,e\right)=\mathbb{O}\left(\nu(1+\epsilon\Lambda)\,e^{\nu(-h\alpha\beta+\alpha f+\beta e+\delta e f)}\,\middle|\,e\,l\,f\right)$, with $\nu=(1+h\,\delta)^{-1}$ and Λ, the "logos", as below.

```
Λ = With[{q = ⅇ^(ν (f α+e β-h α β+e f δ))}, Simplify[q^-1 (2 l ∂_{f,e} q + f ∂_{f,f,e} q + e ∂_{f,e,e} q - 1/2 h ∂_{f,f,e,e} q)]]
```

$-\frac{1}{2}\,\nu\,\Big(-4\,l\,\left(\delta+\alpha\,\beta\,\nu+f\,\alpha\,\delta\,\nu+e\,\beta\,\delta\,\nu+e\,f\,\delta^2\,\nu\right)+\nu\,\Big(e^2\,\beta^2\,\delta\,\nu\,(-2+h\,\delta\,\nu)+2\,e\,\beta\,(2\,\delta+\alpha\,\beta\,\nu)\,(-1+h\,\delta\,\nu)+$
$h\,\left(2\,\delta^2+4\,\alpha\,\beta\,\delta\,\nu+\alpha^2\,\beta^2\,\nu^2\right)+f^2\,\delta\,(\alpha+e\,\delta)\,\nu\,\left(e\,\delta\,(-4+h\,\delta\,\nu)+\alpha\,(-2+h\,\delta\,\nu)\right)+$
$2\,f\,\left(\alpha^2\,\beta\,\nu\,(-1+h\,\delta\,\nu)+e\,\delta^2\,\left(-4+2\,h\,\delta\,\nu+e\,\beta\,\nu\,(-3+h\,\delta\,\nu)\right)+2\,\alpha\,\delta\,\left(-1+h\,\delta\,\nu+e\,\beta\,\nu\,(-2+h\,\delta\,\nu)\right)\right)\Big)\Big)$

(The below is private; don't look)

```
oldΛ = Simplify[
  1/(2 μ^4) (-b α^2 β^2 + u^2 β^2 δ (2 + b δ) + w^2 δ (α + u δ) (α (2 + b δ) + u δ (4 + 3 b δ)) - 4 b α β δ μ + 4 c α β μ^2 - 2 b δ^2 μ^2 +
     4 c δ μ^3 + 2 u β (α β + 2 δ μ (1 + c μ)) + 2 w (α^2 β + 2 α δ (u β (2 + b δ) + μ (1 + c μ)) +
        u δ^2 (u β (3 + 2 b δ) + 2 μ (2 + b δ + c μ)))) /. {μ → ν^-1, b → h, c → l, u → e, w → f}]
```

$\frac{1}{2}\,\nu\,\Big(4\,l\,\left(\delta+\alpha\,\beta\,\nu+f\,\alpha\,\delta\,\nu+e\,\beta\,\delta\,\nu+e\,f\,\delta^2\,\nu\right)+\nu\,\Big(2\,\left(e\,\beta+f\,(\alpha+2\,e\,\delta)\right)\,\nu\,\left(\alpha\,\beta\,\nu+e\,f\,\delta^2\,\nu+\delta\,(2+f\,\alpha\,\nu+e\,\beta\,\nu)\right)+$
$h\,\left(-4\,\alpha\,\beta\,\delta\,\nu-\alpha^2\,\beta^2\,\nu^2+3\,e^2\,f^2\,\delta^4\,\nu^2+4\,e\,f\,\delta^3\,\nu\,(1+f\,\alpha\,\nu+e\,\beta\,\nu)+\delta^2\,\left(-2+f^2\,\alpha^2\,\nu^2+4\,e\,f\,\alpha\,\beta\,\nu^2+e^2\,\beta^2\,\nu^2\right)\right)\Big)\Big)$

```
Simplify[Λ == oldΛ /. ν → (1 + h δ)^-1]
```

True

Implementing $\mathfrak{g}_1$

```
PBWRule = {e → 1, l → 2, f → 3};
B[U@e, U@l] = -U@e; B[U@f, U@l] = U@f; B[U@e, U@f] = h U[] -2 ϵ U[l];
ϵ /: ϵ^2 = 0;
```

```
$TD = 3; ℏ /: ℏ^d_. /; d > $TD := 0;
```

```
x_ ≤ y_ := OrderedQ[{x, y} /. PBWRule]; x_ < y_ := ! OrderedQ[{y, x} /. PBWRule];
Simp[ℰ_] := Collect[ℰ, _U, Expand];
```

```
U_i_[ℰ_] := ℰ /. {h → h_i, t → t_i, u_U ⧴ Replace[u, x_ ⧴ x_i, 1]};
B[U[(x_)_i_], U[(y_)_i_]] := B[U[x_i], U[y_i]] = U_i[B[U@x, U@y]];
B[U[(x_)_i_], U[(y_)_j_]] /; i =!= j := 0;
B[x_, x_] = 0;
B[U[y_], U[x_]] := B[U[y], U[x]] = Simp[-B[U[x], U[y]]];
B[x_, y_] := x ** y - y ** x;
```

```
Unprotect[NonCommutativeMultiply];
NonCommutativeMultiply[x_] := x;
0 ** _ = _ ** 0 = 0;
x_ ** U[] := x; U[] ** x_ := x;
(a_ * x_U) ** (b_ * y_U) := If[a b === 0, 0, Simp[a b (x ** y)]];
(a_ * x_U) ** y_ := Simp[a (x ** y)]; x_ ** (a_ * y_U) := Simp[a (x ** y)];
(x_Plus) ** y_ := (# ** y) & /@ x; x_ ** (y_Plus) := (x ** #) & /@ y;
```

```
U[xx___, x_] ** U[y_, yy___] := If[x ≤ y, U[xx, x, y, yy], U@xx ** (U@y ** U@x + B[U@x, U@y]) ** U@yy];
```

```
UU[l___, x_^n_, r___] := UU[l, Sequence @@ Table[x, {n}], r];
UU[l___, 1, r___] := UU[l, r];
UU[] = U[];
UU[l_, r___] := U[l] ** UU[r];
```

```
UProducts[{}, 0] = {UU[]};
UProducts[{}, n_Integer] /; n > 0 = {};
UProducts[{x_, xs___}, n_Integer] :=
  Sort@Flatten@Table[UU[x^k] ** u, {k, 0, n}, {u, UProducts[{xs}, n - k]}];
UProducts[xs_List, k_Integer, n_Integer] := UProducts[Flatten@Table[x_j, {x, xs}, {j, k}], n];
UProducts[any__, {n_}] := Flatten@Table[UProducts[any, k], {k, 0, n}];
```

```
r_{i_,j_} := Simp[ℏ (h_i UU[1_j] + UU[e_i, f_j])]
```

```
UExp[u_] := Module[{s, t, k},
   s = t = U[]; k = 0;
   While[k < 20 ∧ 0 =!= (t = t ** u), s += t / (++k)!];
   Simp[s]];
R_{i_,j_} := UExp[r_{i,j}];
```

```
m[i_, j_, k_][ℰ_] := Simp[ℰ /. {
    u_U :> UU @@ Join[DeleteCases[u, x_{i|j}], U @@ Cases[u, x_i :> x_k], U @@ Cases[u, x_j :> x_k]],
    h_{i|j} → h_k}]
```

```
𝕆[poly_, specs___] := Module[{vs, us, z},
  vs = Join @@ (First /@ {specs});
  us = Join @@ ({specs} /. (l_ → s_) :> (l /. x_i_ :> x_s));
  Simp@Total[CoefficientRules[Normal@Series[poly, {ℏ, 0, $TD}], vs] /. (p_ → c_) :> c UU @@ (us^p)]
 ]
```

Debt. How does this work?

Verifying the full $\mathfrak{g}_1$ lemma

3. $\mathbb{O}\left(e^{\alpha f+\beta e+\delta e f} \mid f e\right)=\mathbb{O}\left(\nu(1+\epsilon\Lambda)\, e^{\nu(-h\alpha\beta+\alpha f+\beta e+\delta e f)} \mid e\, l\, f\right)$, with $\nu=(1+h\delta)^{-1}$ and Λ, the “logos”, as above.

$TD = 2; $\mathbb{O}\left[e^{\hbar\,(\alpha f_1+\beta e_1+\delta e_1 f_1)},\ \{f_1, e_1\} \to 1\right]$

$$\left(1-\delta\hbar h_1-\alpha\beta\hbar^2 h_1-\delta^2\epsilon\hbar^2 h_1+\delta^2\hbar^2 h_1^2\right) U[\,] + \left(\beta\hbar+2\beta\delta\epsilon\hbar^2-2\beta\delta\hbar^2 h_1\right) U[e_1] + \left(\alpha\hbar+2\alpha\delta\epsilon\hbar^2-2\alpha\delta\hbar^2 h_1\right) U[f_1] + \left(2\delta\epsilon\hbar+2\alpha\beta\epsilon\hbar^2-4\delta^2\epsilon\hbar^2 h_1\right) U[l_1] + \frac{1}{2}\beta^2\hbar^2 U[e_1, e_1] + \left(\delta\hbar+\alpha\beta\hbar^2+4\delta^2\epsilon\hbar^2-2\delta^2\hbar^2 h_1\right) U[e_1, f_1] + 4\beta\delta\epsilon\hbar^2 U[e_1, l_1] + \frac{1}{2}\alpha^2\hbar^2 U[f_1, f_1] + 4\alpha\delta\epsilon\hbar^2 U[l_1, f_1] + \beta\delta\hbar^2 U[e_1, e_1, f_1] + \alpha\delta\hbar^2 U[e_1, f_1, f_1] + 4\delta^2\epsilon\hbar^2 U[e_1, l_1, f_1] + \frac{1}{2}\delta^2\hbar^2 U[e_1, e_1, f_1, f_1]$$

$\Lambda 1 = \Lambda$ /. $\{\alpha \to \hbar\alpha,\ h \to \hbar h_1,\ e \to \hbar e_1,\ l \to l_1,\ f \to f_1\}$

$$-\frac{1}{2}\nu\left(\nu\left(\hbar\left(2\delta^2+4\alpha\beta\delta\nu\hbar+\alpha^2\beta^2\nu^2\hbar^2\right)h_1+\beta^2\delta\nu\hbar^2 e_1^2\left(-2+\delta\nu\hbar h_1\right)+2\beta\hbar\left(2\delta+\alpha\beta\nu\hbar\right)e_1\left(-1+\delta\nu\hbar h_1\right)+\delta\nu\left(\alpha\hbar+\delta\hbar e_1\right)f_1^2\left(\delta\hbar e_1\left(-4+\delta\nu\hbar h_1\right)+\alpha\hbar\left(-2+\delta\nu\hbar h_1\right)\right)+2 f_1\left(\alpha^2\beta\nu\hbar^2\left(-1+\delta\nu\hbar h_1\right)+\delta^2\hbar e_1\left(-4+2\delta\nu\hbar h_1+\beta\nu\hbar e_1\left(-3+\delta\nu\hbar h_1\right)\right)+2\alpha\delta\hbar\left(-1+\delta\nu\hbar h_1+\beta\nu\hbar e_1\left(-2+\delta\nu\hbar h_1\right)\right)\right)\right)-4\left(\delta+\alpha\beta\nu\hbar+\beta\delta\nu\hbar e_1+\alpha\delta\nu\hbar f_1+\delta^2\nu\hbar e_1 f_1\right)l_1\right)$$

$TD = 7;
$\mathbb{O}\left[e^{\hbar\,(\alpha f_1+\beta e_1+\delta e_1 f_1)},\ \{f_1, e_1\} \to 1\right] == \mathbb{O}\left[\nu\left(1+\epsilon\hbar\Lambda 1\right)e^{\hbar\nu\,(-\alpha\beta\hbar h_1+\alpha f_1+\beta e_1+\delta e_1 f_1)}\ /.\ \nu \to \left(1+\hbar h_1\delta\right)^{-1},\ \{e_1, l_1, f_1\} \to 1\right]$

True